Reasoning About Programs

Panagiotis Manolios
Northeastern University

February 22, 2019
Version: 107

1

Introduction

These lecture notes were developed for Logic and Computation, a freshman-level class taught at the College of Computer and Information Science of Northeastern University. Starting in Spring 2008, this is a class that all students in the college are required to take.

The goals of the Logic and Computation course are to provide an introduction to formal logic and its deep connections to computing. Logic is presented from a computational perspective using the ACL2 Sedan theorem proving system. The goal of the course is to introduce fundamental, foundational methods for modeling, designing, specifying and reasoning about computation. The topics covered include propositional logic, recursion, contracts, testing, induction, equational reasoning, termination analysis, term rewriting, and various proof techniques. We show how to use logic to formalize the syntax and semantics of the core ACL2s language, a simple LISP-based language with contracts. We then use the ACL2s language to formally reason about programs, to model systems at various levels of abstraction, to design and specify interfaces between systems and to reason about such composed systems. We also examine decision procedures for fragments of first-order logic and how such decision procedures can be used to analyze models of systems.

The students taking the Logic and Computation class have already taken a programming class in the previous semester, in Racket. The course starts by reviewing some basic programming concepts. The review is useful because at the freshman level students benefit from seeing multiple presentations of key concepts; this helps them to internalize these concepts. For example, in past semesters I have asked students to write very simple programs (such as a program to append two lists together) during the first week of classes and a surprisingly large number of students produce incorrect code.

During the programming review, we introduce the ACL2s language. This is the language we use throughout the semester and it is similar to Racket. The syntax and semantics of the core ACL2s language are presented in a mathematical way. We provide enough information so that students can determine what sequence of glyphs form a well-formed expression and how to formally evaluate well-formed expressions potentially containing user-defined functions with constants as arguments (this is always in a first-order setting). This is a pretty big jump in rigor for students and is advanced material for freshmen students, but they already have great intuitions about evaluation from their previous programming class. This intuition helps them understand the rigorous presentation of the syntax and semantics, which in turns helps strengthen their programming abilities.

The lecture notes are sparse. It would be great to add more exercises, but I have not done that yet. Over the course of many years, we have amassed a large collection of homework problems, so students see lots of exercises, and working through these exercises is a great way for them to absorb the material, but the exercises are not in the notes. You can think of the lecture notes as condensed notes for the course that are appropriate for someone who knows the material as a study guide. The notes can also be used as a starting point by

students, who should mark them up with clarifications as needed when they attend lectures. I advise students to read the lecture notes before class. This way, during class they can focus on the lecture instead of taking notes and they are better prepared to ask for clarifications.

When I started teaching the class, I used the ACL2 book, *Computer-Aided Reasoning, An Approach* by Kaufmann, Manolios and Moore. However, over the years I became convinced that using an untyped first-order logic was not the optimal way of introducing logic and computation to students because they come in with a typed view of the world. That's not to say they have seen type theory; they have not. But, they are surprised when a programming language allows them to subtract a string from a rational number. Therefore, with the help of my Ph.D. student Harsh Chamarthi, I have focused on adding type-like capabilities to ACL2s. Most notably, we added a new data definition framework to ACL2s that supports enumeration, union, product, record, map, (mutually) recursive and custom types, as well as limited forms of parametric polymorphism. We also introduced the `defunc` macro, which allows us to formally specify input and output contracts for functions. These contracts are very general, *e.g.*, we can specify that `/` is given two rationals as input, and that the second rational is not 0, we can specify that `zip` is given two lists of the same length as input and returns a list of the same length as output and so on. Contracts are also checked statically, so ACL2s will not accept a function definition unless it can prove that the function satisfies its contracts and that for every legal input and every possible computation, it is not possible during the evaluation of the function being defined to be in a state where some other function is poised to be evaluated on a value that violates its input contract. I have found that a significant fraction of erroneous programs written by students have contract violations in them, and one of the key things I emphasize is that when writing code, one needs to think carefully about the contracts of the functions used and why the arguments to every function call satisfy the function's contract. Contracts are the first step towards learning how to specify interfaces between systems. With the move to contracts, the ACL2 book became less and less appropriate, which led me to write these notes.

I have distributed these notes to the students in Logic and Computation for several years and they have found lots of typos and have made many suggestions for improvement. Thanks and keep the comments coming!

4

Equational Reasoning

We just finished studying propositional logic, so let's start by considering the following question:

Why do we need more than propositional logic?

After all, we were able to do a lot with propositional logic, including declarative design, security and digital logic.

What we are really after, however, is reasoning about programs, and while propositional logic will play an important role, we need more powerful logics.

To see why, let's simplify things for a moment and consider conjectures involving numbers and arithmetic operations.

Consider the conjecture:

Conjecture 1 $a + b = ba$

What does it mean for this conjecture to be true or false?

Well, there is a source of ambiguity here. If a, b are constants (like 1, 2, etc.) then we can just evaluate the equality and determine if it is true or not.

However, a and b are variables. This is similar to the propositional formulas we saw, *e.g.*,

$$p \wedge q \;\Rightarrow\; p \vee q$$

Recall that p and q are atoms, and the above formula is valid. What that means is that no matter what value p and q have, the above formula is true. Another way of saying this is that the above formula evaluates to true under all assignments.

In Conjecture 1, a and b range over a different domain than the Booleans, let's say they range over the rationals.

So, what we really mean when we say that conjecture 1 is valid (or true) is that for any rational number a and any rational number b, $a + b = ba$. Another way of saying this is that the above formula evaluates to true under all assignments. Notice the similarity with the Boolean case.

Is Conjecture 1 a valid formula?

No. We can come up with a counterexample.

Is Conjecture 1 unsatisfiable?

No. It is both satisfiable and falsifiable. Again, this is exactly the kind of characterization we used to classify Boolean formulas. How many assignments falsify the conjecture? How many assignments satisfy the conjecture?

What about the following conjecture?

Conjecture 2 $a + b = b + a$

Can we come up with a counterexample?

No.

Is the formula valid?

Yes.

How do we prove that a conjecture is valid in the case of propositional logic? We use a truth table, with a row per possible assignment, to show that no counterexample exists. A counterexample is an assignment that evaluates to false.

Can we do something similar here?

Yes, but the number of assignments is unfortunately infinite. That means, we can never completely fill in a table of assignments.

We need a radically new idea here.

We want something that allows us to do a finite amount of work and from that to deduce that there are no counterexamples in the infinite table, were we even able to construct it.

Let's look at how we might do this, but in the context of programs. First we start with the definition of `len`, a function we have already seen.

```
(definec len (l :all) :nat
  (if (consp l)
      (+ 1 (len (cdr l)))
    0))
```

Consider the following conjecture:

Conjecture 3 `(equal (len (list x)) (len x))`

Is this conjecture true (valid) or false (falsifiable)?

What does it mean for Conjecture 3 to be true? That no matter what object of the ACL2s universe `x` is, the above equality holds.

Conjecture 3 is false. Why?

Suppose that `x` = `1`, then the conjecture evaluates to `nil`, *i.e.*,

$$[\![\texttt{(equal (len (list 1)) (len 1))}]\!] = [\![\texttt{(equal 1 0)}]\!] = \texttt{nil}$$

So, finding a counterexample is "easy." All we have to do is to find an assignment under which the conjecture evaluates to `nil`. This is just like the Boolean logic case.

Is Conjecture 3 unsatisfiable? No. Again, all we have to do is find one satisfying assignment, *e.g.*, `x` = `(1)`. How many assignments falsify the conjecture? How many assignments satisfy the conjecture?

What about:

Conjecture 4
```
(equal (len (cons x (list z)))
       (len (cons y (list z))))
```

Here we can't find a counterexample.

How can we go about proving that Conjecture 4 is valid?

```
    (len (cons x (list z)))
= { Def len, instantiation }
    (if (consp (cons x (list z))) (+ 1 (len (cdr (cons x (list z))))) 0)
= { car-cdr axioms }
    (if (consp (cons x (list z))) (+ 1 (len (list z))) 0)
= { consp axioms }
    (if t (+ 1 (len (list z))) 0)
= { if axioms }
    (+ 1 (len (list z)))
```

What we have shown so far is:

$$\texttt{(len (cons x (list z)))} = \texttt{(+ 1 (len (list z)))} \tag{4.1}$$

which we will feel free to write as

$$\texttt{(len (cons x (list z)))} = 1 + \texttt{(len (list z))} \tag{4.2}$$

because it should be clear how to go from (4.1) to (4.2) and because we have been trained to use infix for arithmetic operators since elementary school.

You should be able to continue the proof to show that

$$\texttt{(len (cons x (list z)))} = 2 \tag{4.3}$$

Finish the proof.

Once the proof is done, we have shown that (4.3) is valid. Any validity that we establish via proof is called a *theorem*, so (4.3) is a theorem. To reason about built-in functions such as `consp`, `if`, and `equal` we use *axioms*, which you can think of as built-in theorems providing the semantics of the built-in functions. Every time we define a function that ACL2s admits, we also get a *definitional axiom*, which, for now, you can think of as an axiom stating that the function is equal to its body (but more on this soon). We can then reason from these basic axioms (which are also theorems) using what is called *first order logic*. First order logic includes propositional reasoning, but extends it significantly. We will introduce as much of first order reasoning as needed.

Back to our proof. We are not done with the proof of Conjecture 4, but there are at least two reasonable ways to proceed.

First, we might say:

If we simplify the RHS (Right Hand Side), we get

```
  (len (cons y (list z)))
= { Def len, instantiation }
    ...
= { if axioms }
    (+ 1 (len (list z)))
= { Def len, instantiation }
    ...
= { Arithmetic }
```

2

So, the LHS (Left Hand Side) and RHS are equal.

What we realized is that the same steps that we used to simplify the LHS can be used in a symmetric way to simplify the RHS. In this class we will avoid proofs involving "...". Here's a better way to make the argument:

First, note that (4.3) is a theorem. By instantiating (4.3) with the substitution `((x y))`, we get:

$$\texttt{(len (cons y (list z)))} = 2 \tag{4.4}$$

Putting (4.3) and (4.4) together, we have

$$\texttt{(len (cons x (list z)))} = \texttt{(len (cons y (list z)))}$$

So, Conjecture 4 is a theorem.

We already saw that instantiation can be used in propositional logic. Its use is indispensable when reasoning about ACL2s programs!

This example highlights the new tool we have that allows us to reason about programs: proof. The game we will be playing is to construct proofs of conjectures involving some of the basic functions we have already defined (*e.g.*, `len`, `aapp`, and `rrev`). We will focus on these simple functions because their simplicity allows us to focus exclusively on how to prove theorems without the added complexity of having to understand what conjectures mean.

Once we prove that a conjecture is valid, we say that the conjecture is a *theorem*. We are then free to use that theorem in proving other theorems. This is similar to what happens when we program: we define functions and then we use them to define other functions (*e.g.*, we define `rrev` using `aapp`).

What's new here?

Well, we are beyond the realm of the propositional. We have variables ranging over the ACL2s universe, equality, and functions.

Let's look at equality. To simplify notation, we tend to write expressions involving `equal` using = instead. This is similar to what we did with arithmetic. For example, instead of writing the more technically correct

```
(equal (len (cons x z)) (len (cons y z)))
```

we usually write the more familiar

$$\texttt{(len (cons x z))} = \texttt{(len (cons y z))}$$

We feel free to go back and forth without justification.

If we want to be pedantic, here is how `equal` and = are related.

- $x = y \;\Rightarrow\; \texttt{(equal x y)} = \texttt{t}$
- $x \neq y \;\Rightarrow\; \texttt{(equal x y)} = \texttt{nil}$

When we use = or $\neq$ in expressions, they bind more tightly than any of the propositional operators.

How can we reason about equality? We will use just two properties of equality. First, equality is what is called an *equivalence relation*, *i.e.*, it satisfies the following properties.

- *Reflexivity*: $x = x$

- *Symmetry*: $x = y \;\Rightarrow\; y = x$
- *Transitivity*: $x = y \wedge y = z \;\Rightarrow\; x = z$

That $=$ is an equivalence relation is what allows us to chain together the sequence of equalities in the proof of Conjecture 4 above to conclude that `(len (cons x (list z)))` $= 2$.

The second property of equality we will use is the *Equality Axiom Schema for Functions*: For every function symbol `f` of arity n we have the axiom

$$(x_1 = y_1 \wedge \cdots \wedge x_n = y_n) \;\Rightarrow\; \texttt{(f } x_1 \cdots x_n\texttt{)} = \texttt{(f } y_1 \cdots y_n\texttt{)}$$

To reason about constants, we can use evaluation, *e.g.*, all of the following are theorems.

$$\texttt{t} \neq \texttt{nil}$$

$$\texttt{()} = \texttt{nil}$$

$$1 \neq 2$$

$$2/4 = 4/8$$

$$\texttt{(cons 1 ())} = \texttt{(list 1)}$$

To reason about built-in functions, such as `cons`, `car`, and `cdr`, we have axioms for each of these functions that are derived from their semantics.

$$\texttt{(car (cons x y))} = \texttt{x}$$

$$\texttt{(cdr (cons x y))} = \texttt{y}$$

$$\texttt{(consp (cons x y))} = \texttt{t}$$

$$x = \texttt{nil} \Rightarrow \texttt{(if x y z)} = \texttt{z}$$

$$x \neq \texttt{nil} \Rightarrow \texttt{(if x y z)} = \texttt{y}$$

What about instantiation? It is a rule of inference:
Instantiation: Derive $\varphi|_\sigma$ from φ. That is, if φ is a theorem and σ is a substitution, then by instantiation, $\varphi|_\sigma$ is a theorem.

For example, since this is a theorem:

```
(equal (car (cons x y)) x)
```

we can conclude that the following is a theorem by instantiation.

```
(equal (car (cons (foo x) (bar z))) (foo x))
```

More carefully, a substitution is just a list of the form:

$$\texttt{((}var_1\ term_1\texttt{)} \ \ldots\ \texttt{(}var_n\ term_n\texttt{))}$$

where the var_i are variables (symbols such as `x`, `y`, etc.) that we refer to as *target variables* and the $term_i$ are expressions that we refer to as *images*. We require that the var_i are distinct. The application of this substitution to a formula uniformly replaces every free occurrence of a target variable by its image.

The definition of a free occurrence of a variable will be provided with the help of the following examples. The first point is that there is a distinction between function symbols and variable symbols. Consider:

$$\texttt{(f (g f g) 'f)}|_{\texttt{((f x) (x g) (g f))}} = \texttt{(f (g x f) 'f)}$$

Notice that we only substitute an occurrence of a symbol that corresponds to a variable, which is why the occurrence of `f` in function position `(f ...)` was not changed. On the other hand the occurrence of `f` in `(g f g)` is a variable and it was replaced by its image in the substitution. Also, `'f` is not a variable. It is a constant and it is not affected by any substitution. The general rule is that quoted objects are never modified by a substitution. One more thing to notice with this example is that substitutions are similar to `let`s: you do not keep recursively applying the substitution. For example the free occurrence of `g` became `f`, but we do not recursively keep applying the substitution, which would wind up leading to an infinite loop. If you want an algorithmic way of thinking about substitution, then just start copying the expression until you get to a free occurrence of a target variable and replace it by its image and then continue on with the rest of the expression. Once you get to the end of the expression, you are done. This part is exactly the same as substitutions in propositional logic, something we have already covered and seen many times.

Another point involves bound occurrences of variables. Consider:

$$\texttt{(list x y (let ((y z) (z y)) (list x y z)))}|_{\texttt{((x a) (y b) (z c))}}$$

$$= \texttt{(list a b (let ((y c) (z b)) (list a y z)))}$$

Notice that the `y` in `(let ((y z)) ...)` is not replaced because the variables bound in a `let` are not free. However the `z` in `(let ((y z)) ...)` *is* free, so that gets replaced. Also, since `let` is a parallel binding (review `let`), the second `y` in `(let ((y z) (z y)) ...)` is free. The `y` in the body of the `let` is also *not* free. A `let*` form is equivalent to a nested `let` form, so we do not have to consider it as a special case.

Why is the definition of ACL2s substitution more complicated that it was in the case of propositional logic? Because the ACL2s logic is more powerful and because we want instantiation to be a valide rule of inference, as we use all the time. If we did not make distinctions between free and bound occurrences of variables, then instantiation would not work, *e.g.*, consider:

$$\varphi = \texttt{(implies (natp y) (equal (let ((x 0)) (+ x y)) y))}, \sigma = \texttt{((x 1))}$$

Clearly φ is a theorem, as is:

$$\varphi|_\sigma = \texttt{(implies (natp y) (equal (let ((x 0)) (+ x y)) y))}$$

But, if we were to replace every occurence of `x` with `1`, we would get the following, which is not even an expression, let alone a theorem.

$$\varphi|_\sigma = \texttt{(implies (natp y) (equal (let ((1 0)) (+ 1 y)) y))}$$

If we just replace the second occurrence of `x`, we get the following, which is an expression, but not a theorem.

$$\varphi|_\sigma = \texttt{(implies (natp y) (equal (let ((x 0)) (+ 1 y)) y))}$$

To see why we distinguish between function symbols and variable symbols, consider the theorem

$$\varphi = \texttt{(consp (cons x y))}, \sigma = \texttt{((consp atom))}$$

Clearly, φ is a theorem as is

$$\varphi|_\sigma = \texttt{(consp (cons x y))}$$

But, if we were to replace every occurence of `consp` with `atom`, we would get the following, which is not a theorem.

$$\texttt{(atom (cons x y))}$$

Finally, to see why quoted objects are not variables, consider:

$$\varphi = \texttt{(symbolp 'x)}, \sigma = \texttt{((x 1))}$$

Clearly φ is a theorem, as is:

$$\varphi|_\sigma = \texttt{(symbolp 'x)}$$

But, if we were to replace every occurence of `x` with `1`, we would get the following, which is not a theorem.

$$\texttt{(symbolp '1)}$$

Notice that substitutions only replace variables. You cannot replace expressions or constants. To see why we have this restrictions, consider:

$$\varphi = \texttt{(equal (+ 1 1) 2)}, \sigma = \texttt{((1 0))}$$

Clearly φ is a theorem, but, if we were to replace every occurence of `1` with `0`, we would get the following, which is not a theorem.

$$\varphi = \texttt{(equal (+ 0 0) 2)}$$

Next consider:

$$\varphi = \texttt{(implies (natp x) (equal (+ x x) (* 2 x)))}, \sigma = \texttt{(((+ x x) x))}$$

Clearly φ is a theorem, but, if we were to replace every occurence of `(+ x x)` with `x`, we would get the following, which is not a theorem.

$$\varphi = \texttt{(implies (natp x) (equal x (* 2 x)))}$$

What does it mean to say that the following is a theorem?

$$\texttt{(len (cons x (list z)))} = \texttt{(len (cons y (list z)))}$$

That no matter what you replace `x`, `y`, and `z` with from the ACL2s universe, the LHS and RHS evaluate to the same thing.

Let's try to prove another conjecture.

Conjecture 5 `(aapp (cons x y) z)` = `(cons x (aapp y z))`

```
    (aapp (cons x y) z)
```
= { Def `aapp`, instantiation }
```
    (if (endp (cons x y)) z (cons (first (cons x y)) (aapp (rest (cons x y)) z)))
```
= { Def `endp`, `consp` axioms }
```
    (if nil z (cons (first (cons x y)) (aapp (rest (cons x y)) z)))
```
= { `if` axioms }
```
    (cons (first (cons x y)) (aapp (rest (cons x y)) z))
```
= { `car-cdr` axioms }
```
    (cons x (aapp y z))
```

Unfortunately, the above "proof" has a problem. Unlike `len`, which is defined for the whole ACL2s universe, `aapp` is only defined for true lists.

Recall the definitions:

```
(definec true-listp (l :all) :bool
  (if (consp l)
      (true-listp (cdr l))
    (equal l () )))

(definec endp (l :list) :bool
  (atom l))

(definec aapp (x :tl y :tl) :tl
  ; Aapp appends two lists together
  (if (endp x)
      y
    (cons (first x) (aapp (rest x) y))))
```

Recall that `tlp` is abbreviation for `true-listp`, so we will be using the shorter `tlp` from now on. The definition of functions such as `aapp` give rise to *definitional axioms*. Here is the definitional axiom that `aapp` gives rise to:

```
(tlp x) ∧ (tlp y)
⇒
(aapp x y)
=
(if (endp x)
    y
  (cons (first x) (aapp (rest x) y)))
```

In general, every time we successfully admit a function, we get two theorems of the form

$$ic \;\Rightarrow\; (\texttt{f}\ x_1 ... x_n) = body$$

$$ic \;\Rightarrow\; oc$$

where ic is the input contract for `f`, and where oc is the output contract for `f`. We will be very precise about what "successfully admit" means, but, for now, take this to mean that ACL2s accepts your function definition with `defunc`. When using `definec`, we get similar

theorems. Recall that this involves proving termination, proving the function contracts, and proving the body contracts.

So, we can't expand the definition of `aapp` in the proof of Conjecture 5, unless we know:

$$\texttt{(tlp (cons x y))} \wedge \texttt{(tlp z)}$$

which is equivalent to:

$$\texttt{(tlp y)} \wedge \texttt{(tlp z)}$$

So, what we really proved was:

Theorem 4.1 $\texttt{(tlp y)} \wedge \texttt{(tlp z)} \Rightarrow \texttt{(aapp (cons x y) z)} = \texttt{(cons x (aapp y z))}$

When we write out proofs, we will not explicitly mention input contracts when using a function definition because the understanding is that every time we use a definitional axiom to expand a function, we have to check that we satisfy the input contract, so we don't need to remind the reader of our proof that we did something we all understand always needs doing.

It is often the case that when we think about conjectures that we expect to be valid, we often forget to carefully specify the hypotheses under which they are valid. These hypotheses depend on the input contracts of the functions mentioned in the conjectures, so get into the habit of looking at conjectures and making sure that they have the needed hypotheses. *Contract checking* is the process of checking that a conjecture has all the hypotheses required by the contracts of the functions appearing in the conjecture. *Contract completion* is the process of adding the missing hypotheses (if any) identified during contract checking. Contract checking and completion is similar to what you do when you write functions: you check the body contracts of the functions you define and if you are calling the functions on arguments of the wrong type, then you modify your code appropriately. In the case of function definitions, as we have seen, it is often the case that if the function definition is wrong, there is also a contract violation. Similarly, if a conjecture is not valid, it is often the case that there is a contract violation.

Let's look at another example:

Conjecture 6 $\texttt{(endp x)} \Rightarrow \texttt{(aapp (aapp x y) z)} = \texttt{(aapp x (aapp y z))}$

Can I prove this? Check the contracts of the conjecture.

Contract checking and completion gives rise to:

Conjecture 7 $\texttt{(tlp x)} \wedge \texttt{(tlp y)} \wedge \texttt{(tlp z)} \wedge \texttt{(endp x)} \Rightarrow \texttt{(aapp (aapp x y) z)} = \texttt{(aapp x (aapp y z))}$

By the way, notice all of the hypotheses. Notice the Boolean structure. This is why we studied Boolean logic first! Almost everything we will prove will include an implication.

Notice that in ACL2s, we would technically write:

```
(implies (and (tlp x)
              (tlp y)
              (tlp z)
              (endp x))
         (equal (aapp (aapp x y) z)
```

```
(aapp x (aapp y z))))
```

The first thing to do when proving theorems is to take the Boolean structure into account by writing the conjecture in the form:

$$hyp_1 \wedge hyp_2 \wedge \cdots \wedge hyp_n \;\Rightarrow\; conc$$

where we have as many *hyp*s as possible. We will call the set of top-level hypotheses (*i.e.*, $\{hyp_1, hyp_2, \ldots, hyp_n\}$) our *context*.

Our context for Conjecture 7 is:

C1. `(tlp x)`

C2. `(tlp y)`

C3. `(tlp z)`

C4. `(endp x)`

We then look at our context and see what obvious things our context implies. The obvious thing here is that C1 and C4 imply that `x` must be `nil`, so we extend our context with a *derived context*:

D1. `x` = `nil` { C1, C4 }

Notice that any new facts we add must come with a justification. We will use the convention that all elements of our context will be given a label of the form Ci, where i is a positive integer and that all elements of our derived context will be given a label of the form Di, where i is a positive integer.

The next thing we do is to start with the LHS of the conclusion and to try and reduce it to the RHS, using our proof format. If we need to refer to the context in one of proof step justifications, say Derived Context 1, we write D1.

```
  (aapp (aapp x y) z)
= { Def aapp, D1, Def endp, if axioms }
  (aapp y z)
= { Def aapp, D1, Def endp, if axioms }
  (aapp x (aapp y z))
```

Notice that we took bigger steps than before. Before we might have written:

```
  (aapp (aapp x y) z)
= { Def aapp }
  (aapp (if (endp x) y (cons (first x) (aapp (rest x) y))) z)
= { D1 }
  (aapp (if (endp nil) y (cons (first nil) (aapp (rest nil) y))) z)
= { Def endp }
  (aapp (if t y (cons (first nil) (aapp (rest nil) y))) z)
= { If axioms }
  (aapp y z)
```

...

So, the above four steps were compressed into one step. Why? Because many of the steps we take involve expanding the definition of a function. Function definitions tend to have a top-level `if` or `cond` and as a general rule we will not expand the definition of such a function unless we can determine which case of the top-level `if`-structure will be true. If we just blindly expand function definitions, we'll wind up with a sequence of increasingly complicated terms that don't get us anywhere. So, if we know which case of the top-level `if` is true, then why go to the trouble of writing out the whole body of the function? Why not just write out that one case? Well, that's why we allow ourselves to expand definitions as in the first proof of Conjecture 7.

One other comment about the first proof of Conjecture 7. Students often have no difficulty with the first step, but have difficulty with the second step. The second step requires one to see that the simple term:

```
(aapp y z)
```

can be transformed into the RHS

```
(aapp x (aapp y z))
```

This may seem like a strange thing to do because students are used to thinking about computation as unfolding over time. So, if `x` is `nil` then of course the following holds.

```
  (aapp (aapp x y) z)
```
= { Def `aapp`, ... }
```
  (aapp y z)
```

Because when we compute `(aapp x y)` we get `y`.

What students initially have difficulty with is seeing that you can reverse the flow of time and everything still works. For example the following is true,

```
  (aapp y z)
```
= { Def `aapp`, ... }
```
  (aapp (aapp x y) z)
```

Because starting with `(aapp y z)` we can run time in reverse to get `(aapp x (aapp y z))` (recall `x` is `nil`). In fact, this is "obvious" from the equality (=) axioms that tell us that equality is an equivalence relation (reflexive, symmetric, and transitive). The symmetry axiom tells us that we can view computation as moving forward in time or backward. It just doesn't make a difference.

As an aside, it turns out that in physics, we can't reverse time and so this symmetry we have with computation is not a symmetry we have in our universe. One reason why we can't reverse time in physics is that the second law of thermodynamics precludes it. The second law of thermodynamics implies that entropy increases over time. There is an even more fundamental reason why time is not reversible. This second reason has to do with the fundamental laws of physics at the quantum level, whereas the second law of thermodynamics is thought to be a result of the initial conditions of our universe. The second reason is that in our current understanding of the universe, there are very small violations of time reversibility exhibited by subatomic particles. The extent of the violations is not fully understood and probably has something to do with the imbalance of matter and antimatter in the visible universe. There is almost no antimatter in the visible universe and one of the big open problems in physics is trying to understand why that is the case.

4.1 Testing Conjectures

Recall that since Conjecture 7 is a theorem, whatever we replace the free variables with, the conjecture will evaluate to `t`. A convenient way of checking the conjecture using ACL2s is to use `let`, as follows:

```
(let ((x nil)
      (y nil)
      (z nil))
  (implies (and (tlp x)
                (tlp y)
                (tlp z)
                (endp x))
           (equal (aapp (aapp x y) z)
                  (aapp x (aapp y z)))))
```

An even more convenient method is to use ACL2s to test the conjecture. Here is how:

```
(test?
  (implies (and (tlp x)
                (tlp y)
                (tlp z)
                (endp x))
           (equal (aapp (aapp x y) z)
                  (aapp x (aapp y z)))))
```

There are three possible outcomes.

1. ACL2s proves that the conjecture is a theorem.
2. ACL2s finds a counterexample, *i.e.*, the conjecture is falsifiable.
3. None of the above hold, *i.e.*, the conjecture satisfies all of the tests ACL2s tries.

Consider another example. Consider the claim that `aapp2`, below, is equivalent to `aapp`.

```
(definec aapp2 (x :tl y :tl) :tl
  (if (endp y)
      x
    (cons (first y) (aapp2 x (rest y)))))
```

The claim is false, but `aapp2` works fine on many tests, *e.g.*,

```
(check= (aapp2 '(1 2) '(1 2)) '(1 2 1 2))
(check= (aapp2 nil nil) nil)
(check= (aapp2 nil '(1 2 3)) '(1 2 3))
(check= (aapp2 '(1 2 3) nil) '(1 2 3))
```

Here is how we can use ACL2s to test the conjecture that `aapp2` is equivalent to our definition.

```
(test?
  (implies (and (tlp x) (tlp y))
           (equal (aapp2 x y)
```

```
                (aapp x y))))
```

ACL2s gives us counterexamples. It also shows us cases in which the conjecture is true.

Now, suppose that the specification for `aapp2` only stated that `(aapp2 x y)` must return a list that contains all of the elements in `x` and all of the elements in `y`, where order doesn't matter, but repetitions do. The definition of `aapp2` above satisfies the specification. In addition, there are many semantically different functions that satisfy the specification. How can we write tests that are independent of the implementation? We cannot write simple `check=`'s because there are exponentially many correct answers that `aapp2` could return. We can't test that `aapp2` is equal to our solution for the same reason. But, we can write conjectures that capture the specification and ACL2s can be used to test these conjectures.

Here is one way of doing this. We test that every element in `(aapp2 x y)` is also an element of `(aapp x y)` and conversely.

```
; check that if a is in aapp2, it is in aapp
(test?
  (implies (and (tlp x)
                (tlp y)
                (in a (aapp2 x y)))
           (in a (aapp x y))))

; check that if a is in aapp, it is in aapp2
(test?
  (implies (and (tlp x)
                (tlp y)
                (in a (aapp x y)))
           (in a (aapp2 x y))))
```

Exercise 4.1 *Unfortunately, we can define* `aapp2` *in a way that does not satisfy the specification, but does satisfy the above* `test?`*'s. Exhibit such a definition and check that it passes the above tests. A better solution is to test that* `aapp2` *is a permutation of* `aapp`*. Define a function that checks if its arguments are permutations of one another and use this to test both your faulty definition of* `aapp2` *and the definition given above.*

You can control how much testing ACL2s does. The default number of tests depends on the mode, but you can set it to whatever number you want, *e.g.*, here is how to instruct ACL2s to run 1,000 tests.

```
(acl2s-defaults :set num-trials 1000)
```

To summarize, ACL2s provides `test?`, a powerful facility for automatically testing programs. Instead of having to manually write tests, ACL2s generates as many tests as requested automatically. The other major advantage is that we do not have to specify exactly what functions have to do. In the `aapp2` example above, we did not have to say what `aapp2` returns; instead, we specified the properties we expect `aapp2` to satisfy. The advantage is that we *decouple* the testing of `aapp2` from the development of `aapp2`. In fact, even if we change the implementation of `aapp2`, the tests can remain the same.

4.2 Equational Reasoning with Complex Propositional Structure

Many of the conjectures we will examine have rich propositional structure. We now examine how to reason about such conjectures.

Conjecture 8

```
(consp x)
⇒
 [[(tlp (rest x)) ∧ (tlp y) ∧ (tlp z)
   ⇒
   (aapp (aapp (rest x) y) z) = (aapp (rest x) (aapp y z))]
  ⇒
  [(tlp x) ∧ (tlp y) ∧ (tlp z)
   ⇒
   (aapp (aapp x y) z) = (aapp x (aapp y z))]]
```

The above conjecture has the form

$$A \Rightarrow [B \Rightarrow C]$$

where

A is `(consp x)`

B is `[(tlp (rest x)) ∧ (tlp y) ∧ (tlp z)`
`⇒ (aapp (aapp (rest x) y) z) = (aapp (rest x) (aapp y z))]`

C is `[(tlp x) ∧ (tlp y) ∧ (tlp z)`
`⇒ (aapp (aapp x y) z) = (aapp x (aapp y z))]`

What we are doing here is identifying some of the propositional structure of Conjecture 8. Here's why. Remember *exportation*, a propositional validity we have already encountered.

$$A \Rightarrow [B \Rightarrow C] \equiv [A \wedge B] \Rightarrow C$$

We will use exportation almost all the time. We now use it to rewrite Conjecture 8 so that the context has as many conjunctions as possible. After applying exportation to Conjecture 8, we get:

```
[(consp x) ∧
 [(tlp (rest x)) ∧ (tlp y) ∧ (tlp z)
  ⇒
  (aapp (aapp (rest x) y) z) = (aapp (rest x) (aapp y z))]
⇒
[(tlp x) ∧ (tlp y) ∧ (tlp z)
   ⇒
   (aapp (aapp x y) z) = (aapp x (aapp y z))]]
```

Applying exportation again and rearranging conjuncts gives us:

Conjecture 9

```
[(consp x) ∧
 (tlp x) ∧
 (tlp y) ∧
 (tlp z) ∧
 [(tlp (rest x)) ∧ (tlp y) ∧ (tlp z)
  ⇒
  (aapp (aapp (rest x) y) z) = (aapp (rest x) (aapp y z))]]
⇒
  (aapp (aapp x y) z) = (aapp x (aapp y z))]]
```

Now, we can extract the context. Doing so gives us:

C1. `(consp x)`

C2. `(tlp x)`

C3. `(tlp y)`

C4. `(tlp z)`

C5. `[(tlp (rest x)) ∧ (tlp y) ∧ (tlp z)] ⇒`
`[(aapp (aapp (rest x) y) z) = (aapp (rest x) (aapp y z))]`

Notice that we *cannot* use exportation on C5 to add the hypotheses of C5 to our context. Why?

We will be confronted with implications in our context (like C5) over and over. Usually what we will need is the consequent of the implication, but we can only use the consequent if we can also establish the antecedent, so we will try to do that in the derived context. Here's how:

D1. `(tlp (rest x))` { Def `tlp`, C2, C1 }

D2. `(aapp (aapp (rest x) y) z) = (aapp (rest x) (aapp y z))` { C5, D1, C3, C4, MP }

So, notice what we did. First we added D1 to our derived context. How did we get D1? Well, we know `(tlp x)` (C2) and `(consp x)` (C1) so if we use the definitional axiom of `tlp`, we get D1: `(tlp (rest x))`.

Now, we have extended our context to include the antecedent of D1, so by propositional logic (*Modus Ponens*, abbreviated MP), we get that the conclusion also holds, *i.e.*, D2.

Recall that Modus Ponens tells us that if the following two formulas hold

$$A \Rightarrow B$$

$$A$$

Then so does the formula

$$B$$

We are now ready to prove the theorem. We start with the LHS of the equality in the conclusion of Conjecture 9.

```
  (aapp (aapp x y) z)
```
= { Def `aapp`, C1, C2, C3 }
```
  (aapp (cons (first x) (aapp (rest x) y)) z)
```
= { Theorem 4.1 }
```
  (cons (first x) (aapp (aapp (rest x) y) z))
```
= { D2 }
```
  (cons (first x) (aapp (rest x) (aapp y z)))
```
= { Def `aapp`, C1, C2, C3, C4 }
```
  (aapp x (aapp y z))
```

4.3 The difference between theorems and context

It is very important to understand the difference between a formula that is a theorem and one that appears in a context. A formula that appears in a context cannot be instantiated. It can only be used as is, in the proof attempt for the conjecture from which it was extracted. This is a major difference. Our contexts will never include theorems we already know. Theorems we already know are independent of any conjecture we are trying to prove and therefore do not belong in a context. A context is always formula specific.

Here is an example that shows why instantiation of context formulas leads to unsoundness. Here is a "proof" of

$$\texttt{x} = 1 \;\Rightarrow\; 0 = 1 \tag{4.5}$$

Context:

C1. $\texttt{x} = 1$

Proof

```
  0
```
= { Instantiate C1 with `((x 0))` }
```
  1
```

So, now we have a "proof" of (4.5), but using (4.5) we can get:

$$\texttt{nil} \tag{4.6}$$

How?
Instantiate (4.5) with `((x 1))`, use Propositional logic, and Arithmetic.
Now we have a proof for any conjecture we want, *e.g.*,

$$\varphi \tag{4.7}$$

How?
Well, `nil` (false) implies anything, so this is a theorem

$$\texttt{nil} \Rightarrow \varphi$$

Now, φ follows using (4.6) and Modus Ponens.

The point is that a context, including the derived context, is *completely* different from a theorem. The context of (4.5) does not tell us that for all `x`, `x` = 1. It just tells us that `x` = 1 in the context of conjecture (4.5). Contexts are just a mechanism for extracting propositional structure from a conjecture, which in turn allows us to focus on the important part of a proof and to minimize the writing we have to do.

4.4 Undecidability of Equational Reasoning

In the cases we have seen so far, it was easy to decide if a conjecture was true or false, and with a good amount of testing, we would have identified the false conjectures. In fact, ACL2s does that automatically.

Is this always the case?

No.

Consider Fermat's last theorem.

Conjecture 10 *For all positive integers x, y, z, and n, where $n > 2$, $x^n + y^n \neq z^n$.*

In 1637, Fermat wrote about the above:

> "I have a truly marvelous proof of this proposition which this margin is too narrow to contain."

This is called Fermat's Last Theorem. It took 357 years for a correct proof to be found (by Andrew Wiles in 1995).

We can use Fermat's last theorem to construct a conjecture that is hard to prove in ACL2s. We start with the following definition.

```
(defunc f (x y z n)
  :input-contract (and (posp x)
                       (posp y)
                       (posp z)
                       (natp n)
                       (> n 2))
  :output-contract (booleanp (f x y z n))
  (not (equal (+ (expt x n) (expt y n))
              (expt z n))))
```

Now consider the following conjecture.

```
(thm (implies (and (posp x)
                   (posp y)
                   (posp z)
                   (natp n)
                   (> n 2))
              (f x y z n)))
```

So, proving theorems may be hard.

Notice also that if the output contract was

```
:output-contract (equal (f x y z n) t)
```

then ACL2s would have to prove a theorem that eluded mankind for centuries in order to even admit `f`!

But, it is easy to find a counterexample to a conjecture that is not a theorem, right?

That is not true either. There are many examples of conjectures that took a long time to resolve, and which turned out to be false.

In fact, the satisfiability problem for arithmetic expressions over the integers, using only $=, +, \cdot$ is *undecidable*. That means that no algorithm exists that given such an arithmetic expression returns "yes" if there is an assignment that makes the expression true and "no" otherwise. Notice that this also means that the validity problem is undecidable because φ is satisfiable iff $\neg\varphi$ is not valid, *i.e.*, if we had a decision procedure for validity, we could use it to obtain a decision procedure for satisfiability. We will see a proof of a classic undecidability result (the undecidability of the halting problem) in a subsequent chapter.

We end by showing that even admitting `definec` functions in ACL2s is no easier than proving the validity of formulas. Consider any conjecture φ over variables $x_1, \ldots, x_n$. Now consider the following ACL2s code.

```
(defdata true t)
(definec f (x_1 :all ··· x_n :all) :true
  φ)
```

ACL2s has to prove φ to admit `f` because the function contract is:

```
(implies (and (allp x_1) ... (allp x_n)) (truep φ))
```

but since `(allp x)` = `t`, this is equivalent to:

```
(truep φ)
```

4.5 Arithmetic

We can also reason about arithmetic functions. For example, consider the following conjecture

$$\sum_{i=0}^{n} i = \frac{n(n+1)}{2}$$

That is, summing up $0, 1, \ldots, n$ gives $\frac{n(n+1)}{2}$.

We can prove this using mathematical induction.

Here is how we do it in ACL2s. First, we have to define Σ.

```
(definec sum (n :nat) :nat
  (if (equal n 0)
      0
    (+ n (sum (- n 1)))))
```

We can prove that `(sum n)` = $\frac{\texttt{n}(\texttt{n}+1)}{2}$, which is formalized as:

```
(implies (natp n)
```

```
(equal (sum n)
       (/ (* n (+ n 1)) 2)))
```

by mathematical induction. How?
First, we have the base case.

$$\texttt{(natp n)} \wedge \texttt{(equal n 0)} \Rightarrow \texttt{(sum n)} = \texttt{(/ (* n n+1) 2)} \tag{4.8}$$

Second, we have the induction step.

$$\begin{aligned}&\texttt{(natp n)} \wedge \texttt{n} \neq 0 \wedge (\texttt{(natp n-1)} \Rightarrow \texttt{(sum n-1)} = \texttt{(/ (* n-1 n) 2)}) \\ &\Rightarrow \texttt{(sum n)} = \texttt{(/ (* n n+1) 2)}\end{aligned} \tag{4.9}$$

Here is the proof, starting with (4.8).
Context:

C1. `(natp n)`

C2. `(equal n 0)`

Proof:

```
    (sum n)
= { Def sum, C2 }
    0
= { Arithmetic, C2 }
    (/ (* n n+1) 2)
```

Here is the proof of (4.9).
Context:

C1. `(natp n)`

C2. $\texttt{n} \neq 0$

C3. `(natp n-1) ⇒ (sum n-1) = (/ (* n-1 n) 2)`

Derived Context:

D1. `(natp n-1)` { C1, C2 }

D2. `(sum n-1) = (/ (* n-1 n) 2)` { C3, D1, MP }

Proof:

```
    (sum n)
```

$=$ { Def `sum`, C2 }
$\quad n$ + `(sum n-1)`
$=$ { D2 }
$\quad n$ + `(/ (* n-1 n) 2)`
$=$ { Arithmetic }
$\quad (2n + n(n-1))/2$
$=$ { Arithmetic }
$\quad (2n + n^2 - n)/2$
$=$ { Arithmetic }
$\quad (n^2 + n)/2$
$=$ { Arithmetic }
$\quad n(n+1)/2$

4.6 How to prove theorems

When presented with a conjecture, make sure that you check contracts, as shown above.

If the contracts checking succeeds, make sure you understand what the conjecture is saying.

Once you do, see if you can find a counterexample.

If you can't find a counterexample, try to prove that the conjecture is a theorem.

One often iterates over the last two steps.

During the proof process, you have available to you all the theorems we have proved so far. This includes all of the axioms (`car-cdr` axioms, `if` axioms, ...), all the definitional axioms (Def `aapp`, `len`, ...), all the contract theorems (Contract of `aapp`, `len`, ...). These theorems can be used at any time in any proof and can be instantiated using any substitution. They are a great weapon that will help you prove theorems, so make sure you understand the set of already proven theorems.

There are also local facts extracted from the conjecture under consideration. Recall that the first step is to try and rewrite the conjecture into the form:

$$[C_1 \wedge C_2 \wedge \ldots \wedge C_n] \Rightarrow \text{RHS}$$

where we try to make RHS as simple as possible. $C_1, \ldots, C_n$ are going to be the first n components of our context. Formulas in the context are specific to the conjecture under consideration. They are completely different from theorems (as per the above discussion). A good amount of manipulation of the conjecture may be required to extract the maximal context, but it is well worth it.

The next step is to see what other facts the $C_1, \ldots, C_n$ imply. For example, if the current context is:
Context:

C1. `(endp x)`

C2. `(tlp x)`

then, we create the derived context and populate it as follows.

D1. `x` = `nil` { C1, C2 }

This will happen a lot. Another case that will happen a lot is:

C1. `(consp x)`

C2. `(tlp x)`

C3. `(tlp (rest x))` $\Rightarrow \varphi$

then the the derived context includes:
Derived Context:

D1. `(tlp (rest x))` { C1, C2, Def `tlp` }

D2. φ { C3, D1, MP }

where MP is Modus Ponens.

As was the case when we studied propositional logic, we have "word problems," something we now consider:

Conjecture 11 $x \leq xy$ *if* $y \geq 1$

Question: What does the above conjecture mean, anyway?

It means that for any values of x, and y, if $y \geq 1$ then $x \leq xy$.

Really? Any values? What if x and y are functions or strings or ...? Usually the domain is implicit, *i.e.*, "clear from context."

We will be using ACL2s, and we can't appeal to "context." This is a good thing!

Notice also that we can use ACL2s, a programming language, to make mathematical statements. Of course! Programming languages are mathematical objects and you reason about programs the way you reason about the natural numbers, the reals, sets, etc.: you prove theorems.

In ACL2s, we have to be precise about the conditions under which we expect the conjecture to hold. The conjecture can be formalized in ACL2s as follows:

```
(thm
 (implies (and (rationalp x)
               (rationalp y)
               (>= y 1))
          (<= x (* x y))))
```

In standard mathematical notation it is:

$$\langle \forall x, y \in Q :: y \geq 1 \ \Rightarrow \ x \leq xy \rangle$$

Is the above conjecture true?

Well, when given a conjecture, we can try one of two things:

1. Try to falsify it.
2. Try to prove it is correct.

How do we falsify a conjecture?

Simple exhibit a counterexample.

Remember that in the design recipe, we construct examples and tests. You should do the same thing with conjectures. That is, we can test that the conjecture is true on examples. Here are some:

1. $x = 0, y = 0$
2. $x = 12, y = 1/3$
3. $x = 9, y = 3/2$

Any others?

How do we test this in ACL2s? Put the conjecture in the body of a let.

```
(let ((x 0)
      (y 0))
  (implies (and (rationalp x)
                (rationalp y)
                (>= y 1))
           (<= x (* x y))))
```

We are using a programming language, so we can do better. We can write a program to test the conjecture on a large number of cases. How many cases are there? We can use a random number generator to "randomly" sample from the domain. We'll see how to do that in ACL2s.

```
(test?
  (implies (and (rationalp x)
                (rationalp y)
                (>= y 1))
           (<= x (* x y))))
```

If all of the tests pass, then we can try to prove that the conjecture is a theorem.

What would a "proof" of the above conjecture look like?

Most proofs are informal and it takes a long time for students to understand what constitutes an informal proof. This happens by osmosis over time.

In our case, we have a simple rule: it's a proof if ACL2s says it is.

```
(thm
 (implies (and (rationalp x)
               (rationalp y)
               (>= y 1))
          (<= x (* x y))))
```

Of course, this isn't a theorem.

Let's consider another example:

Conjecture 12 *x(y+z) = xy + xz*

How do we write this in ACL2s?

```
(thm (implies (and (rationalp x)
                   (rationalp y)
```

```
                (rationalp z))
           (equal (* x (+ y z))
                  (+ (* x y) (* x z)))))
```

Is the above conjecture true?

Well, we can try to falsify it.

```
(let ((x 0)
      (y 0)
      (z 0))
  (equal (* x (+ y z))
         (+ (* x y) (* x z))))
```

We can try many examples. We can automatically generate random examples.

```
(test?
 (implies (and (rationalp x)
               (rationalp y)
               (rationalp z))
          (equal (* x (+ y z))
                 (+ (* x y) (* x z)))))
```

When do we give up falsifying this?

Can we just try all the possibilities? If we had infinite time. Do we? Maybe (ask a physicist), but, as a practical matter, we currently don't.

Maybe we should consider a proof. Can we prove the above?

One answer might be: "of course, multiplication distributes over addition."

In ACL2s, the conjecture turns out to be true

```
(thm (implies (and (rationalp x)
                   (rationalp y)
                   (rationalp z))
              (equal (* x (+ y z))
                     (+ (* x y) (* x z)))))
```

This is pretty amazing because a proof gives us a finite way of running an infinite number of examples. That's the power of logic and mathematics.

When ACL2s proves this theorem, is it thinking?

> The question of whether Machines Can Think ... is about as relevant as the question of whether Submarines Can Swim.
>
> Edsger W. Dijkstra: EWD898, 1984

See the EWD archives at the University of Texas at Austin.

Here is another example

```
(definec aapp (a :tl b :tl) :tl
  (if (endp a)
      b
    (cons (first a) (aapp (rest a) b))))

(definec rrev (x :tl) :tl
  (if (endp x)
```

```
      nil
    (aapp (rrev (rest x)) (list (first x)))))

(definec in (a :all X :tl) :bool
  (and (consp X)
       (or (equal a (first X))
           (in a (rest X)))))

(definec del (a :all X :tl) :tl
  (cond ((endp X) nil)
        ((equal a (car X)) (cdr X))
        (t (cons (car X) (del a (cdr X))))))
```

Conjecture 13

```
(tlp x)
⇒
(in a x)   ⇒   (not (in a (del a x)))
```

Using induction (something we will describe later), the above conjecture leads to the following proof obligation:

```
(and (implies (tlp x)
              (implies (endp x)
                       (implies (in a x)
                                (not (in a (del a x))))))
     (implies (tlp x)
              (implies (and (consp x)
                            (equal a (first x)))
                       (implies (in a x)
                                (not (in a (del a x))))))
     (implies
      (tlp x)
      (implies
       (and (consp x)
            (not (equal a (first x)))
            (implies (tlp (rest x))
                     (implies (in a (rest x))
                              (not (in a (del a (rest x)))))))
       (implies (in a x)
                (not (in a (del a x)))))))
```

Is this true? If so, give a proof. Is it false? If so, exhibit a counterexample.

Try this before reading further.

Conjecture 13 is false, *e.g.*, consider

```
(let ((x '(1 1))
      (a 1))
  ...)
```

where ... in the above let is Conjecture 13.

What about the following conjecture?

Conjecture 14

```
(tlp x)
⇒
(in a x)   ⇒   (in a (aapp x y))
```

Which by induction leads to the following proof obligation:

```
(and (implies (tlp x)
              (implies (endp x)
                       (implies (in a x)
                                (in a (aapp x y)))))
     (implies (tlp x)
              (implies (and (consp x)
                            (equal a (first x)))
                       (implies (in a x)
                                (in a (aapp x y)))))
     (implies
      (tlp x)
      (implies
       (and (consp x)
            (not (equal a (first x)))
            (implies (tlp (rest x))
                     (implies (in a (rest x))
                              (in a (aapp (rest x) y)))))
       (implies (in a x)
                (in a (aapp x y))))))
```

Is this true? If so, give a proof. Is it false? Is so, exhibit a counterexample.

Try this before reading further.

This conjecture fails contract checking. After contract completion, it is true and you should be able to prove it by breaking Conjecture 14 into three parts and proving each in turn.

Proofs of the three parts appear in Figures 4.1, 4.2 and 4.3. These proofs highlight the proof format we use when writing proofs on a computer. If we are writing proofs with paper and pencil, we will follow a similar proof format, but we will typically not explicitly write the exportation, contract completion and goal steps.

The proof of the first part, in Figure 4.1, shows that it is possible for us to prove a theorem in the context, if we can derive `nil`, because `nil` implies anything. The "QED" indicates the end of a proof and comes from the Latin *Quod Erat Demonstrandum*, which means "that which was to be demonstrated."

The proof of the second part, in Figure 4.2, shows that the Derived Context section of the proof is optional. In fact, if no exportation is possible, the Exportation section is optional. If no contract completion is needed, then the Contract Completion section is optional. If we derive `nil`, even the Goal and Proof sections are optional.

The proof of the third part, in Figure 4.3 includes all possible sections that we currently have available (more sections will be introduced later). This proof also provides an example

Conjecture 14-part-1:

```
(implies (tlp x)
         (implies (endp x)
                  (implies (in a x)
                           (in a (aapp x y)))))
```

Exportation:

```
(implies (and (tlp x)
              (endp x)
              (in a x))
         (in a (aapp x y)))
```

Contract Completion:

```
(implies (and (tlp x)
              (tlp y)
              (endp x)
              (in a x))
         (in a (aapp x y)))
```

Context:

```
C1. (tlp x)
C2. (tlp y)
C3. (endp x)
C4. (in a x)
```

Derived Context:

```
D1. (equal x nil) { C1, C3, Def tlp }
D2. nil { Def in, C4, D1 }
```

QED

Figure 4.1: Proof format example using only context

of nested implications. Notice that the fourth top-level hypothesis in the Exportation section has to remain an implication because exportation does not allow us to take its hypotheses and make them top-level hypotheses. Make sure you understand this! However, we did use exportation to simplify the fourth top-level hypothesis, so always apply exportation as much as possible everywhere you can, not just at the top level. During the exportation step, we allow any propositional simplification, as long as the resulting formula does not allow any further exportation simplifications.

When we have nested implications, as we do in C5, one of the goals when constructing the derived context is to get our hands of the consequent of such implications. The proof in Figure 4.3 shows an example of that. The idea is to establish all the hypotheses of C5 (D1 and D2 in our example) and to then use Modus Ponens (MP) to obtain the conclusion (D3). The conclusion is typically what we need in the proof.

Exercise 4.2 *Use* `definec` *to define a function* `del-all` *that given* `a` *of type* `:all` *and* `X` *of type* `:tl` *deletes all occurrences of* `a` *from* `X`.

Prove the following conjectures (which will require contract completion).

```
Conjecture del-all-1:
(implies (endp x) (not (in a (del-all a x))))
```

Conjecture 14-part-2:

```
(implies (tlp x)
         (implies (and (consp x)
                       (equal a (first x)))
                  (implies (in a x)
                           (in a (aapp x y)))))
```

Exportation:

```
(implies (and (tlp x)
              (consp x)
              (equal a (first x))
              (in a x))
         (in a (aapp x y)))
```

Contract Completion:

```
(implies (and (tlp x)
              (tlp y)
              (consp x)
              (equal a (first x))
              (in a x))
         (in a (aapp x y)))
```

Context:

```
C1. (tlp x)
C2. (tlp y)
C3. (consp x)
C4. (equal a (first x))
C5. (in a x)
```

Goal: `(in a (aapp x y))`

Proof:

```
     (in a (aapp x y))
```

= { Def `aapp`, C1, C3 }

```
     (in a (cons (first x) (aapp (rest x) y)))
```

= { Def `in`, `car-cdr` axioms, C3 }

```
     (or (equal a (first x)) (in a (aapp (rest x) y)))
```

= { C4, PL }

```
     t
```

QED

Figure 4.2: Proof format with no derived context

Conjecture 14-part-3:

```
(implies (tlp x)
         (implies (and (consp x)
                       (not (equal a (first x)))
                       (implies (tlp (rest x))
                                (implies (in a (rest x))
                                         (in a (aapp (rest x) y)))))
                  (implies (in a x)
                           (in a (aapp x y)))))
```

Exportation:

```
(implies (and (tlp x)
              (consp x)
              (not (equal a (first x)))
              (implies (and (tlp (rest x))
                            (in a (rest x)))
                       (in a (aapp (rest x) y)))
              (in a x))
         (in a (aapp x y)))
```

Contract Completion:

```
(implies (and (tlp x)
              (tlp y)
              (consp x)
              (not (equal a (first x)))
              (implies (and (tlp (rest x))
                            (in a (rest x)))
                       (in a (aapp (rest x) y)))
              (in a x))
         (in a (aapp x y)))
```

Context:

```
C1. (tlp x)
C2. (tlp y)
C3. (consp x)
C4. (not (equal a (first x)))
C5. (implies (and (tlp (rest x)) (in a (rest x)))
             (in a (aapp (rest x) y)))
C6. (in a x)
```

Derived Context:

```
D1. (tlp (rest x)) { Def tlp, C1, C3 }
D2. (in a (rest x)) { Def in, C6, C3, C4 }
D3. (in a (aapp (rest x) y)) { C5, D1, D2, MP }
```

Goal: `(in a (aapp x y))`

Proof:

```
     (in a (aapp x y))
= {  Def aapp, C1, C3  }
     (in a (cons (first x) (aapp (rest x) y)))
= {  Def in, car-cdr axioms, C3  }
     (or (equal a (first x)) (in a (aapp (rest x) y)))
= {  D3, PL  }
     t
```

QED

Figure 4.3: Proof format with nested implications

```
Conjecture del-all-2:
(implies
 (consp x)
 (implies
  (equal a (first x))
  (implies (implies
    (tlp (rest x))
    (not (in a (del-all a (rest x)))))
   (not (in a (del-all a x))))))

Conjecture del-all-3:
(implies
 (consp x)
 (implies
  (not (equal a (first x)))
  (implies (implies
    (tlp (rest x))
    (not (in a (del-all a (rest x)))))
   (not (in a (del-all a x))))))
```

Consider the definition of `nodups`, a function that checks is a true-list has no duplicate elements.

```
(definec nodups (l :tl) :bool
  (or (endp l)
      (and (not (in (first l) (rest l)))
           (nodups (rest l)))))
```

The function `num-unique` determines how many unique elements a true-list contains.

```
(definec num-unique (l :tl) :nat
  (cond ((endp l) 0)
        ((in (first l) (rest l))
         (num-unique (rest l)))
        (t (+ 1 (num-unique (rest l))))))
```

Exercise 4.3 *Prove the following claim. Use propositional logic to break the claim into cases.*

```
(implies (tlp l)
         (implies (or (endp l)
                      (and (not (endp l))
                           (in (first l) (rest l))
                           (implies (tlp (rest l))
                                    (<= (num-unique (rest l))
                                        (llen (rest l)))))
                      (and (not (endp l))
                           (not (in (first l) (rest l)))
                           (implies (tlp (rest l))
                                    (<= (num-unique (rest l))
```

```
                                    (llen (rest l))))))
                (<= (num-unique l)
                    (llen l))))
```

Exercise 4.4 *Prove the following claim. Use propositional logic to break the claim into cases.*

```
(implies (and (tlp l)
              (nodups l))
         (and (implies (endp l)
                       (equal (num-unique l) (llen l)))
              (implies (and (not (endp l))
                            (implies (and (tlp (rest l))
                                          (nodups (rest l)))
                                     (equal (num-unique (rest l))
                                            (llen (rest l)))))
                       (equal (num-unique l) (llen l)))))
```

Exercise 4.5 *You are given the following lemma.*

```
(implies (tlp l)
         (<= (num-unique l) (llen l)))
```

Prove the following claim. Use propositional logic to break the claim into cases.

```
(and
 (implies (and (tlp l) (endp l))
          (equal (equal (num-unique l) (llen l))
                 (nodups l)))
 (implies (and (tlp l)
               (not (endp l))
               (in (car l) (cdr l))
               (implies (tlp (cdr l))
                        (equal (equal (num-unique (cdr l))
                                      (llen (cdr l)))
                               (nodups (cdr l)))))
          (equal (equal (num-unique l) (llen l))
                 (nodups l)))
 (implies (and (tlp l)
               (not (endp l))
               (not (in (car l) (cdr l)))
               (implies (tlp (cdr l))
                        (equal (equal (num-unique (cdr l))
                                      (llen (cdr l)))
                               (nodups (cdr l)))))
          (equal (equal (num-unique l) (llen l))
                 (nodups l))))
```

Exercise 4.6 *You are given the following lemma.*

```
(implies (and (tlp x) (tlp y))
          (equal (in a (aapp x y))
                 (or (in a x) (in a y))))
```

Prove the following claim. Use propositional logic to break the claim into cases.

```
(and
 (implies (and (tlp x) (tlp y))
          (implies (endp x)
                   (<= (num-unique (aapp x y))
                       (+ (num-unique x) (num-unique y)))))
 (implies (and (tlp x) (tlp y))
          (implies (consp x)
                   (implies
                    (implies (tlp (cdr x))
                             (<= (num-unique (aapp (cdr x) y))
                                 (+ (num-unique (cdr x)) (num-unique y))))
                    (<= (num-unique (aapp x y))
                        (+ (num-unique x) (num-unique y)))))))
```

Exercise 4.7 *You are given the following lemma.*

```
(implies (and (tlp x) (tlp y))
          (equal (in a (aapp x y))
                 (or (in a x) (in a y))))
```

Prove the following claim. Use propositional logic to break the claim into cases.

```
(and
 (implies (and (tlp x) (endp x))
          (equal (in a (rrev x)) (in a x)))
 (implies (and (tlp x)
               (not (endp x))
               (implies (tlp (cdr x))
                        (equal (in a (rrev (cdr x)))
                               (in a (cdr x)))))
          (equal (in a (rrev x)) (in a x))))
```

Exercise 4.8 *You are given the following lemmas.*

```
(implies (and (tlp x)
              (tlp y))
         (equal (num-unique (aapp x y))
                (num-unique (aapp y x))))
(implies (tlp x)
         (equal (in a (rrev x))
                (in a x)))
```

Prove the following claim. Use propositional logic to break the claim into cases.

```
(and
```

```
(implies (and (tlp x) (endp x))
         (equal (num-unique (rrev x))
                (num-unique x)))
(implies (tlp x)
         (implies (and (consp x)
                       (in (car x) (cdr x))
                       (implies (tlp (cdr x))
                                (equal (num-unique (rrev (cdr x)))
                                       (num-unique (cdr x)))))
                  (equal (num-unique (rrev x))
                         (num-unique x))))
(implies (and (tlp x)
              (consp x)
              (implies (and (not (in (car x) (cdr x)))
                            (implies (tlp (cdr x))
                                     (equal (num-unique (rrev (cdr x)))
                                            (num-unique (cdr x)))))
                       (equal (num-unique (rrev x))
                              (num-unique x)))))
```